# $Id: PKGBUILD 88191 2013-04-13 15:25:43Z svenstaro $
# Maintainer: Sven-Hendrik Haase <sh@lutzhaase.com>

pkgname=csfml
pkgver=2.0
pkgrel=1
pkgdesc='C bindings for sfml'
arch=('i686' 'x86_64')
url='http://www.sfml-dev.org/'
license=('zlib')
depends=("sfml=${pkgver}")
makedepends=('cmake' 'doxygen')
source=("csfml-${pkgver}::https://github.com/LaurentGomila/CSFML/archive/2.0.tar.gz")
md5sums=('d609c9e3b115d480d8c61911d774472c')

build() {
  cd "$srcdir"/CSFML-${pkgver}

  mkdir build && cd build
  cmake .. \
      -DCMAKE_INSTALL_PREFIX=/usr \
      -DBUILD_DOC=true
  make
  make doc
}

package() {
  cd "$srcdir"/CSFML-${pkgver}/build

  make DESTDIR="$pkgdir/" install

  install -Dm644 ../license.txt ${pkgdir}/usr/share/licenses/${pkgname}/LICENSE
}

